


2.2. Gröbner bases at work: Inverse Kinematics in Robotics

We consider robots with prismatic and revolute joints. The kinematics of such robots can be described by multivariate polynomial equations, after having represented angles α by their sines and cosines and having added the equation $\sin^2(\alpha) + \cos^2(\alpha) = 1$ to the set of polynomial equations. W.r.t. the geometry of robots, there are basically two problems to be considered:

- forward kinematics: determines the position (and orientation) of the end-effector for given lengths of prismatic joints and angles of revolute joints
- inverse kinematics: determines possible lengths and angles from a predetermined goal position of the end-effector.

Whereas a forward kinematics problem always has exactly one solution, an inverse kinematics problem could have no, exactly one, or several (possibly infinitely many) solutions.

Example:



This is a family of robots (l_1, l_2 are the parameters of the family) with 2 degrees of freedom.

l_1, l_2	lengths of the two robot arms (prismatic joints)
p_x, p_y, p_z	x -, y -, and z -coordinates of the position of the end-effector
δ_1, δ_2	angles describing the rotations of the revolute joints
s_1, s_2, c_1, c_2	sines and cosines of δ_1, δ_2 , respectively

Let the lengths l_1, l_2 of the arms (prismatic joints) be fixed. This means we are considering one particular robot in this family. Since this robot has 2 degrees of freedom, we can in general prescribe 2 coordinates of a point in space, say the x and z coordinates, and then

the remaining y coordinates is fixed up to finitely many possibilities. Of course, if we specify coordinates out of the reach of the robot, we will get no solutions.

So we consider the following system of algebraic equations:

given l_1, l_2, p_x, p_z ,

solve for s_1, c_1, s_2, c_2, p_y

$$\begin{aligned} l_2 \cdot c_1 \cdot c_2 - p_x &= 0, \\ l_2 \cdot s_1 \cdot c_2 - p_y &= 0, \\ l_2 \cdot s_2 + l_1 - p_z &= 0, \\ c_1^2 + s_1^2 - 1 &= 0, \\ c_2^2 + s_2^2 - 1 &= 0. \end{aligned}$$

These equations can be transformed into a Gröbner basis in the polynomial ring $\mathbb{Q}(l_1, l_2, p_x, p_z)[c_1, c_2, s_1, s_2, p_y]$:

$$\begin{aligned} -l_2^2 + l_1^2 - 2l_1p_z + p_z^2 + p_x^2 + p_y^2 &= 0, \\ l_2s_2 + l_1 - p_z &= 0, \\ (l_2^2 - l_1^2 + 2l_1p_z - p_z^2) \cdot s_1^2 - l_2^2 + l_1^2 - 2l_1p_z \\ &\quad + p_z^2 + p_x^2 = 0, \\ (l_2^3 - l_2l_1^2 + 2l_2l_1p_z - l_2p_z^2 - l_2p_x^2) \cdot c_2 \\ &\quad + (-l_2^2 + l_1^2 - 2l_1p_z + p_z^2) \cdot s_1 \cdot p_y = 0, \\ (l_2^2 - l_1^2 + 2l_1p_z - p_z^2 - p_x^2) \cdot c_1 - p_x \cdot s_1 \cdot p_y &= 0, \end{aligned}$$

In this Gröbner basis the variables “are separated”, i.e. we can solve for 1 variable at a time (starting from the last polynomial up to the first).

So, for instance, for

$$\begin{array}{ll} l_1 = 30 & \dots\dots \text{length of first bar} \\ l_2 = 45 & \dots\dots \text{length of second bar} \\ p_x = \frac{45\sqrt{6}}{4} \simeq 27.5567 & \dots\dots x\text{-coordinate of end-eff.} \\ p_z = \frac{45\sqrt{2}}{2} + 30 \simeq 61.8198 & \dots\dots z\text{-coordinate of end-eff.} \end{array}$$

we get (among others) the solution

$$p_y = \frac{45\sqrt{2}}{4}, \quad s_2 = \frac{\sqrt{2}}{2}, \quad s_1 = \frac{1}{2}, \quad c_2 = \frac{\sqrt{2}}{2}, \quad c_1 = \frac{\sqrt{3}}{2},$$

i.e. the angles have to be set to $\delta_1 = 30^\circ$, $\delta_2 = 45^\circ$.

All this can be computed in Maple 16 as follows:

```
> with(Groebner):
> F := { l2*c1*c2 - px,
>        l2*s1*c2 - py,
>        l2*s2 + l1 - pz,
>        c1^2 + s1^2 - 1,
```

```

>      c2^2 + s2^2 - 1 };
      F := {l2 c1 c2 - px, l2 s1 c2 - py, l2 s2 + l1 - pz, c1^2 + s1^2 - 1, c2^2 + s2^2 - 1}
> G := Basis(F,plex(c1,c2,s1,s2,py))
      G := { -l2^2 + l1^2 - 2l1pz + pz^2 + px^2 + py^2,
              l2s2 + l1 - pz,
              (l2^2 - l1^2 + 2l1pz - pz^2)s1^2 - l2^2 + l1^2 - 2l1pz + pz^2 + px^2,
              (l2^3 - l2l1^2 + 2l2l1pz - l2pz^2 - l2px^2)c2 + (-l2^2 + l1^2 - 2l1pz + pz^2)s1py,
              (l2^2 - l1^2 + 2l1pz - pz^2 - px^2)c1 - pxs1py }
> l1:=30; l2:=45; px:=45*sqrt(6)/4; pz:=45*sqrt(2)/2+30;
      l1 := 30
      l2 := 45
      px := 45/4*sqrt(6)
      pz := 45/2*sqrt(2) + 30
> solve(G[1]);
      45/4*sqrt(2), -45/4*sqrt(2)
> py:= 45*sqrt(2)/4;
      py := 45/4*sqrt(2)
> solve(G[2]);
      1/2*sqrt(2)
> s2:= %;
      s2 := 1/2*sqrt(2)
> solve(G[3]);
      1/2, -1/2
> s1:=1/2;
      s1 := 1/2
> solve(G[4]);
      1/2*sqrt(2)
> c2:= %;
      c2 := 1/2*sqrt(2)
> solve(G[5]);
      1/4*sqrt(12)
> c1:= sqrt(3)/2;
      c1 := 1/2*sqrt(3)

```

So we have solved a particular inverse kinematics problem for this robot

□